



Analisis Simulasi Inverter Multilevel Kaskade dengan Teknik *Selective Harmonic Elimination Pulse Width Modulation*

Ayong Hiendro^{#1}, Ismail Yusuf^{#2}

[#]Teknik Elektro, Universitas Tanjungpura
Jalan Prof. DR. H. Hadari Nawawi, Pontianak, Indonesia

¹ayong.hiendro@ee.untan.ac.id

²ismail.yusuf@ee.untan.ac.id

Abstrak— Salah satu teknik pengendalian saklar-saklar semikonduktor daya dari inverter multilevel adalah menggunakan *Pulse Width Modulation* (PWM). Dari berbagai teknik PWM, *Selective Harmonic Elimination Pulse Width Modulation* (SHEPWM) merupakan teknik yang andal dalam mengeliminasi harmonisa-harmonisa tingkat rendah. Tujuan dari makalah ini adalah menjabarkan proses perhitungan, perancangan dan simulasi inverter multilevel kaskade dengan menggunakan teknik SHEPWM. Pulsa-pulsa pensaklaran setiap saklar semikonduktor daya inverter dibangun berdasarkan sudut-sudut pensaklaran optimum agar komponen harmonisa tertentu dapat tereliminasi. Demikian pula gelombang tegangan AC yang dihasilkan dapat diprediksi bentuknya berdasarkan hasil rancangan. Selanjutnya simulasi komputer dilakukan dengan perangkat lunak PSPICE. Analisis hasil simulasi terhadap inverter 9-level menunjukkan komponen harmonisa ke-3, 5 dan 7 tereliminasi sama sekali. Hasil ini sesuai dengan perhitungan dan perancangan secara teoritis.

Kata kunci— Indeks Modulasi, Inverter Multilevel, *Pulse Width Modulation*, *Selective Harmonic Elimination*, Sudut Pensaklaran Optimum

I. PENDAHULUAN

Seiring meningkatnya kebutuhan industri dan aplikasi pada sistem energi baru dan terbarukan (EBT) terhadap perangkat inverter bermampuan daya listrik besar yang tidak dapat dipenuhi oleh inverter 2-level [1], inverter multilevel [2] menjadi semakin menarik perhatian bagi para peneliti di bidang elektronika daya. Selain itu, di dalam aplikasi berdaya listrik besar, agar rugi-rugi pensaklaran tidak menjadi besar maka diperlukan frekwensi pensaklaran yang rendah dan hal ini mampu dipenuhi oleh inverter multilevel.

Secara umum terdapat tiga jenis topologi inverter multilevel: inverter multilevel kapasitor layang (*flying capacitor multilevel inverter*) [3], inverter multilevel dioda-jepit (*diode-clamp multilevel inverter*) [4], dan inverter multilevel kaskade (*cascade multilevel inverter*) [5]. Di antara semua topologi tersebut, inverter multilevel

kaskade merupakan jenis yang paling banyak dipakai untuk aplikasi pada sistem-sistem EBT yang beroperasi pada tegangan menengah dan tegangan tinggi. Hal ini disebabkan oleh sifat inverter multilevel kaskade yang modular, yang artinya inverter multilevel kaskade dapat dibangun dari modul-modul kecil inverter sesuai dengan kebutuhan [6].

Permasalahan yang berkaitan dengan inverter multilevel adalah bagaimana menentukan dan mengendalikan saklar-saklar semikonduktor daya untuk menghasilkan tegangan listrik AC yang berkualitas baik. Berbagai teknik pengendalian saklar semikonduktor daya telah dikembangkan, yaitu: *Sinusoidal Pulse Width Modulation* (SPWM) [7], *Space Vector Pulse Width Modulation* (SVPWM) [8] dan *Selective Harmonic Elimination Pulse Width Modulation* (SHEPWM).

Berbagai penelitian juga telah dilakukan untuk meningkatkan kinerja inverter multilevel, antara lain: a) eliminasi harmonisa menggunakan prakalkulasi fungsi-fungsi pensaklaran [9], b) mengembangkan topologi-topologi inverter multilevel [10],[11] dan c) memperbaharui strategi kendali [12]. Penelitian lainnya [13]-[15], melaporkan bahwa inverter-inverter multilevel yang sudah dikembangkan berhasil beroperasi pada frekwensi rendah.

Selanjutnya dari hasil penelitian di [16]-[18] telah membuktikan bahwa teknik SHEPWM merupakan teknik yang paling andal digunakan untuk memperbaiki kualitas bentuk gelombang tegangan AC inverter multilevel. Teknik SHEPWM juga dapat digunakan untuk mengurangi besar frekwensi pensaklaran dan menurunkan THD inverter multilevel. Dengan demikian, inverter multilevel dengan teknik SHEPWM dapat dioperasikan pada frekwensi rendah dengan daya listrik besar [19].

Permasalahan utama yang berkaitan dengan teknik SHEPWM adalah bagaimana memperoleh sudut-sudut pensaklaran optimum dan mengaplikasikannya guna mengeliminasi komponen harmonisa tertentu dari

tegangan AC yang dibangkitkan oleh inverter multilevel [20],[21].


Di dalam makalah ini, dijabarkan proses perancangan inverter multilevel kaskade dengan menggunakan teknik SHEPWM untuk mengeliminasi komponen harmonis tertentu dari tegangan AC yang dibangkitkan. Sudut-sudut penyalian optimum yang digunakan diekstrak dari hasil penelitian pada [22]. Selanjutnya analisis dan simulasi komputer dilakukan dengan menggunakan perangkat lunak PSPICE untuk memvalidasi hasil perancangan dan perhitungan secara teoritis.

II. TINJAUAN PUSTAKA


A. Inverter Multilevel Kaskade

Inverter multilevel kaskade tersusun dari N unit inverter satu-fasa dan N unit sumber tegangan DC. Setiap inverter satu-fasa memiliki 4 saklar semikonduktor S_1, S_2, S_3 dan S_4 serta terhubung ke sumber tegangan DC. Untuk sumber tegangan DC seimbang, maka $V_{DC1} = V_{DC2} = \dots = V_{DCN} = V_{DC}$. Inverter-inveter satu-fasa tersebut tersambung secara seri sehingga membentuk inverter multilevel sebagaimana ditunjukkan dalam gambar 1.


Setiap inverter satu-fasa membangkitkan tegangan AC (lihat gambar 2). Selanjutnya penjumlahan tegangan AC yang dibangkitkan oleh masing-masing inverter satu-fasa ($V_{AC1} + V_{AC2} + \dots + V_{ACN}$) akan menghasilkan tegangan AC bertingkat seperti ditunjukkan dalam gambar 3. Dengan melalui proses pensaklaran yang tepat maka inverter multilevel mampu membangkitkan tegangan AC menengah hingga tinggi dari sumber tegangan DC rendah.



Gambar 1 Rangkaian inverter multilevel kaskade



Gambar 2 Rangkaian dan tegangan AC yang dibangkitkan inverter satu-fasa pada sudut-saklar θ



Gambar 3 Tegangan AC bertingkat yang dihasilkan inverter multilevel

B. Teknik Selective Harmonic Elimination Pulse Width Modulation

Persamaan matematika untuk bentuk gelombang tegangan AC yang dibangkitkan oleh inverter multilevel, sebagaimana ditunjukkan dalam gambar 3, dapat direpresentasikan menggunakan deret Fourier [23], sebagai berikut:

$$v(\omega t) = V_1 \sin(\omega t) + V_2 \sin(2\omega t) + \dots + V_N \sin(N\omega t) \quad (1)$$

Karena bentuk gelombang tegangan AC simetris, maka magnitudo komponen harmonis genap ($V_2, V_4, V_6, \dots, V_{2N}$) bernilai nol, sedangkan komponen harmonis ganjil, yaitu: $V_1, V_3, V_5, \dots, V_{(2N-1)}$ dapat dihitung sebagai berikut:

$$V_n = \frac{4V_{DC}}{n\pi} \left[\sum_{k=1}^N \cos(n\theta_k) \right] \quad (2)$$

Selective harmonic elimination pulse-width modulation atau disingkat SHEPWM adalah suatu teknik yang dapat digunakan untuk mengeliminasi komponen harmonis tertentu dari tegangan AC yang dibangkitkan oleh inverter multilevel. Penerapan teknik SHEPWM dapat dilakukan apabila sudut-sudut pensaklaran untuk mengeliminasikan komponen harmonis-harmonis tertentu sudah diketahui. Untuk mengeliminasi komponen harmonis ke-3,5,7, ..., $(2N-1)$, maka dibuat $V_3 = V_5 = V_7 = \dots = V_{(2N-1)} = 0$, sehingga sudut-sudut pensaklaran optimum dapat ditemukan dengan cara menyelesaikan serangkaian persamaan nonlinier sebagai berikut:

$$\begin{aligned} \frac{4}{\pi M} \left[\sum_{k=1}^N \cos(\theta_k) \right] - 1 &= 0 \\ V_3 = \frac{4V_1}{3\pi M} \left[\sum_{k=1}^N \cos(3\theta_k) \right] &= 0 \\ V_5 = \frac{4V_1}{5\pi M} \left[\sum_{k=1}^N \cos(5\theta_k) \right] &= 0 \\ V_{(2N-1)} = \frac{4V_1}{5\pi M} \left[\sum_{k=1}^N \cos((2N-1)\theta_k) \right] &= 0 \quad (3) \end{aligned}$$

Di dalam persamaan (3) terdapat parameter pengendali tegangan AC yang dihasilkan oleh inverter multilevel, yaitu indeks modulasi atau M , yang nilainya adalah rasio tegangan AC fundamental terhadap tegangan DC, atau dapat dituliskan sebagai $M = V_1/V_{DC}$.

III. HASIL DAN PEMBAHASAN


Dengan menggunakan teknik SHEPWM, maka level inverter menentukan komponen harmonis tertentu dari tegangan AC yang akan dieliminasi. Untuk meneliminasi harmonis ke-3 maka dapat mempergunakan inverter 5-level. Inverter 7-level mengeliminasi harmonis ke-3 dan ke-5. Selanjutnya inverter 9-level mengeliminasi harmonis ke 3, 5 dan 7. Semakin tinggi level inverter maka semakin banyak komponen harmonis yang tereliminasi, namun memerlukan jumlah sudut pensaklaran yang lebih banyak pula.

A. Inverter 5-Level

Inverter 5-level terbentuk dari dua inverter satu-fasa yang tersambung secara seri. Rangkaian inverter 5-level ditunjukkan dalam gambar 4.

Supaya menghasilkan tegangan AC dengan harmonis ke-3 tereliminasi maka diperlukan 2 sudut pensaklaran optimum, yaitu θ_1 dan θ_2 . Untuk memperoleh 2 sudut pensaklaran tersebut maka harus menyelesaikan sistem persamaan nonlinier sebagai berikut:

$$\begin{aligned} \frac{4}{\pi M} [\cos(\theta_1) + \cos(\theta_2)] &= 1 \\ \frac{4}{3\pi M} [\cos(3\theta_1) + \cos(3\theta_2)] &= 0 \quad (4) \end{aligned}$$



Gambar 4 Rangkaian inverter 5-level


Hasil penyelesaian sudut-sudut pensaklaran optimum berdasarkan laporan penelitian dalam [22], disajikan dalam Tabel 1.

Sebagai contoh, untuk $M = 1,2$ maka sudut-sudut pensaklaran optimum adalah $\theta_1 = 27,0341^\circ$ dan $\theta_2 = 87,0341^\circ$. Apabila frekwensi tegangan AC yang diinginkan adalah 50Hz, maka periode satu siklus pensaklaran adalah $T = 1/50 = 20\text{ms}$. Selanjutnya waktu pensaklaran dapat dihitung, $t_1 = (27,0341/360) \times 20\text{ms} = 1,5019\text{ms}$ dan $t_2 = (87,0341/360) \times 20\text{ms} = 4,8352\text{ms}$.


TABEL I
SUDUT-SUDUT PENYALAHAN INVERTER 5-LEVEL

M	θ_1	θ_2
1.10300	29.9897	89.9897
1.11000	29.7795	89.7795
1.20000	27.0341	87.0341
1.30000	23.8795	83.8795
1.40000	20.5927	80.5926
1.50000	17.1427	77.1427
1.60000	13.4879	73.4879
1.70000	9.5685	69.5684
1.80000	5.2928	65.2928
1.90000	0.5070	60.5084
1.95000	2.1566	57.8434
2.00000	5.0804	54.9196
2.10000	12.2216	47.7784
2.20000	26.0211	33.9789
2.20500	29.0308	30.9692

Setelah diketahui waktu pensaklaran, maka rancangan pulsa-pulsa pensaklaran untuk inverter 5-level adalah sebagaimana ditunjukkan dalam gambar 5 dan 6.




Gambar 5 Pulsa-pulsa pensaklaran untuk: S₁₁, S₁₂, S₂₁ dan S₂₂



Gambar 6 Pulsa-pulsa pensaklaran untuk: S₁₃, S₁₄, S₂₃ dan S₂₄


Apabila $V_{DC} = 100$ Volt maka tegangan AC yang dihasilkan oleh inverter 5-level pada $M = 1,2$ adalah seperti ditunjukkan dalam gambar 7.



Gambar 7 Tegangan AC yang dihasilkan inverter 5-level

B. Inverter 7-Level

Tiga inverter satu-fasa membentuk inverter 7-level seperti ditunjukkan dalam gambar 8.



Gambar 8 Rangkaian inverter 7-level

Inverter 7-level memerlukan 3 sudut pensaklaran optimum untuk mengeliminasi harmonisa ke-3 dan 5. Sistem persamaan nonlinier untuk memperoleh θ_1 , θ_2 dan θ_3 adalah

$$\begin{aligned} \frac{4}{\pi M} [\cos(\theta_1) + \cos(\theta_2) + \cos(\theta_3)] &= 1 \\ \frac{4}{3\pi M} [\cos(3\theta_1) + \cos(3\theta_2) + \cos(3\theta_3)] &= 0 \\ \frac{4}{5\pi M} [\cos(5\theta_1) + \cos(5\theta_2) + \cos(5\theta_3)] &= 0 \end{aligned} \quad (5)$$


Sudut-sudut pensaklaran optimum [22] disajikan dalam Tabel 2.

Sebagai contoh, untuk $M = 2,1$ maka sudut-sudut pensaklaran optimum adalah $\theta_1 = 11,9849^\circ$, $\theta_2 = 47,9205^\circ$ dan $\theta_3 = 89,9443^\circ$. Untuk frekwensi tegangan AC 50Hz, maka waktu pensaklaran adalah $t_1 = (11,9849/360) \times 20ms = 0,6658ms$, $t_2 = (47,9205/360) \times 20ms = 2,6623ms$ dan $t_3 = (89,9443/360) \times 20ms = 4,9969ms$.

TABEL III
SUDUT-SUDUT PENYALAN INVERTER 7-LEVEL


M	θ_1	θ_2	θ_3
2.09740	11.9998	47.9994	89.9996
2.10000	11.9849	47.9205	89.9443
2.20000	11.6983	44.8209	87.7458
2.30000	12.0709	41.5484	85.4039
2.40000	13.2264	38.0001	82.9074
2.50000	15.3230	33.9581	80.2355
2.60000	19.0907	28.5832	77.3550
2.62000	20.4599	26.9223	76.7497
2.63778	23.5211	23.6085	76.2025
-	-	-	-
3.06364	19.9694	20.1314	58.1703
3.09000	12.2670	26.5350	56.2858
3.10000	10.2314	27.8482	55.4971
3.12000	5.1131	30.3335	53.7492
3.12730	0.4077	31.2520	53.0364
3.12734	0.1756	31.2569	53.0325

Setelah diketahui waktu pensaklaran, maka rancangan pulsa-pulsa pensaklaran untuk inverter 7-level adalah sebagaimana ditunjukkan dalam gambar 9 dan 10.




Gambar 9 Pulsa-pulsa pensaklaran untuk: S₁₁, S₁₂, S₂₁, S₃₁, S₂₂ dan S₃₂

Apabila V_{DC} = 100 Volt maka tegangan AC yang dihasilkan oleh inverter 7-level pada M = 2,1 adalah seperti ditunjukkan dalam gambar 11



Gambar 10 Pulsa-pulsa pensaklaran untuk: S₁₃, S₁₄, S₂₃, S₃₃, S₂₄ dan S₃₄



Gambar 11 Tegangan AC yang dihasilkan inverter 7-level

C. Inverter 9-Level

Inverter 9-level dibangun oleh 4 inverter satu-fasa yang tersusun secara seri. Inverter 9-level memerlukan 4 sudut pensaklaran optimum untuk mengeliminasi harmonisa ke-3, 5 dan 7.

Sistem persamaan nonlinier untuk memperoleh θ_1 , θ_2 , θ_3 dan θ_4 adalah

$$\begin{aligned} \frac{4}{\pi M} [\cos(\theta_1) + \cos(\theta_2) + \cos(\theta_3) + \cos(\theta_4)] &= 1 \\ \frac{4}{3\pi M} [\cos(3\theta_1) + \cos(3\theta_2) + \cos(3\theta_3) + \cos(3\theta_4)] &= 0 \\ \frac{4}{5\pi M} [\cos(5\theta_1) + \cos(5\theta_2) + \cos(5\theta_3) + \cos(5\theta_4)] &= 0 \\ \frac{4}{7\pi M} [\cos(7\theta_1) + \cos(7\theta_2) + \cos(7\theta_3) + \cos(7\theta_4)] &= 0 \end{aligned} \quad (6)$$

Sudut-sudut pensaklaran optimum [22] disajikan dalam Tabel 3.

TABEL IIII
SUDUT-SUDUT PENYALAHAN INVERTER 9-LEVEL

M	θ_1	θ_2	θ_3	θ_4
3.093000	11.6703	26.9363	56.0555	89.9995
3.094000	11.6664	26.9269	56.0279	89.9824
3.096000	11.6585	26.9083	55.9728	89.9481
3.100000	11.6419	26.8720	55.8625	89.8793
3.120000	11.5423	26.7068	55.3068	89.5322
3.200000	10.8169	26.3546	53.0106	88.0910
3.280000	9.41940	26.5741	50.5585	86.5634
3.320000	8.39160	26.9191	49.2508	85.7659
3.400000	5.25370	28.1202	46.3875	84.0986
3.440000	2.03270	29.0199	44.7651	83.2263
3.446000	0.73590	29.1763	44.5053	83.0931
3.446800	0.24520	29.1978	44.4703	83.0753
3.446900	0.00720	29.2004	44.4658	83.0730
-	-	-	-	-
4.089435	0.0563	24.9097	35.1107	60.8681

M	θ_1	θ_2	θ_3	θ_4
4.090000	1.6703	24.7115	35.2312	60.8269
4.100000	8.2859	21.0034	37.0578	60.0672
4.105000	11.4165	18.3527	37.8488	59.6636
4.107000	13.6124	16.3062	38.1531	59.4970
4.107350	14.6905	15.2524	38.2059	59.4676
4.107370	14.9626	14.9813	38.2082	59.4662

D. Simulasi PSPICE


Untuk menghasilkan tegangan AC dari inverter 9-level, sebagai contoh parameter-parameter yang digunakan adalah: $V_{DC} = 100$ Volt, $M = 3,2$, dan $f = 50$ Hz ($T = 20$ ms). Dari Tabel 3, sudut-sudut penyalaan optimum untuk $M = 3,2$ adalah: $\theta_1 = 10.8169^\circ$, $\theta_2 = 26.3546^\circ$, $\theta_3 = 53.0106^\circ$, dan $\theta_4 = 88.0910^\circ$.

Kode-kode PSPICE untuk simulasi inverter 9-level dengan menggunakan parameter-parameter tersebut adalah:


```
*INVERTER SHEPWM 9 LEVEL
.PARAM Freq=50 ; * Parameters: Freq = output frequency
Vs1 1 1a 100V ; DC input voltage
Vs2 4 4a 100V ; DC input voltage
Vs3 6 6a 100V ; DC input voltage
Vs4 8 8a 100V ; DC input voltage
*
Rb11 11 12 50
Vg11 12 3 PULSE(0 10 0.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
Rb12 17 18 50
Vg12 18 1a PULSE(0 10 0.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
Rb13 13 14 50
Vg13 14 2 PULSE(0 10 10.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
Rb14 15 16 50
Vg14 16 1a PULSE(0 10 10.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
*
Rb21 19 20 50
Vg21 20 2 PULSE(0 10 1.4641ms 1ns 1ns 7.0717ms {1/{Freq}})
*
VG22a 22a 0 PULSE(0 10 0.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
VG22b 22b 0 PULSE(0 10 10.6009ms 1ns 1ns 0.8632ms {1/{Freq}})
VG22c 22c 0 PULSE(0 10 18.539ms 1ns 1ns 0.8632ms {1/{Freq}})
Eq220 220 0 VALUE(V(22a)+V(22b)+V(22c))
Rb22 25 26 50
Eg22 26 4a value{V(220)}
*
Rb23 21 22 50
Vg23 22 5 PULSE(0 10 11.4641ms 1ns 1ns 7.0717ms {1/{Freq}})
*
VG24a 24a 0 PULSE(0 10 0.6009ms 1ns 1ns 0.8632ms {1/{Freq}})
VG24b 24b 0 PULSE(0 10 8.5359ms 1ns 1ns 0.8632ms {1/{Freq}})
VG24c 24c 0 PULSE(0 10 10.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
Eq240 240 0 VALUE(V(24a)+V(24b)+V(24c))
Rb24 23 24 50
Eg24 24 4a value{V(240)}
*
Rb31 27 28 50
Vg31 28 5 PULSE(0 10 2.9450ms 1ns 1ns 4.1099ms {1/{Freq}})
*
VG32a 32a 0 PULSE(0 10 0.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
VG32b 32b 0 PULSE(0 10 10.6009ms 1ns 1ns 2.3441ms {1/{Freq}})
VG32c 32c 0 PULSE(0 10 17.0550ms 1ns 1ns 2.3441ms {1/{Freq}})
Eq320 320 0 VALUE(V(32a)+V(32b)+V(32c))
Rb32 33 34 50
Eg32 34 6a value{V(320)}
*
Rb33 29 30 50
Vg33 30 7 PULSE(0 10 12.9450ms 1ns 1ns 4.1099ms {1/{Freq}})
*
VG34a 34a 0 PULSE(0 10 0.6009ms 1ns 1ns 2.3441ms {1/{Freq}})
VG34b 34b 0 PULSE(0 10 7.0550ms 1ns 1ns 2.3441ms {1/{Freq}})
VG34c 34c 0 PULSE(0 10 10.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
Eq340 340 0 VALUE(V(34a)+V(34b)+V(34c))
Rb34 31 32 50
Eg34 32 6a value{V(340)}
*
Rb41 35 36 50
Vg41 36 7 PULSE(0 10 4.8939ms 1ns 1ns 0.2121ms {1/{Freq}})
*
VG42a 42a 0 PULSE(0 10 0.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
VG42b 42b 0 PULSE(0 10 10.6009ms 1ns 1ns 4.2930ms {1/{Freq}})
VG42c 42c 0 PULSE(0 10 15.1061ms 1ns 1ns 4.2930ms {1/{Freq}})
Eq420 420 0 VALUE(V(42a)+V(42b)+V(42c))
Rb42 41 42 50
Eg42 42 8a value{V(420)}
*
```

```
Rb43 37 38 50
Vg43 38 0 PULSE(0 10 14.8939ms 1ns 1ns 0.2121ms {1/{Freq}})
*
VG44a 44a 0 PULSE(0 10 0.6009ms 1ns 1ns 4.2930ms {1/{Freq}})
VG44b 44b 0 PULSE(0 10 5.1061ms 1ns 1ns 4.2930ms {1/{Freq}})
VG44c 44c 0 PULSE(0 10 10.6009ms 1ns 1ns 8.7981ms {1/{Freq}})
Eq440 440 0 VALUE(V(44a)+V(44b)+V(44c))
Rb44 39 40 50
Eg44 40 8a value{V(440)}
^
D11 3 1 DMD ; Diodes with model DMD
D12 1a 2 DMD
D13 2 1 DMD
D14 1a 3 DMD
*
D21 2 4 DMD
D22 4a 5 DMD
D23 5 4 DMD
D24 4a 2 DMD
*
D31 5 6 DMD
D32 6a 7 DMD
D33 7 6 DMD
D34 6a 5 DMD
*
D41 7 8 DMD
D42 8a 0 DMD
D43 0 8 DMD
D44 8a 7 DMD
*
.MODEL DMD D (IS=2.2E-15 BV=1800V TT=0) ;Diode model parameters
Q11 1 11 3 QMOD ;BJTs with model QMOD
Q12 2 17 1a QMOD
Q13 1 13 2 QMOD
Q14 3 15 1a QMOD
*
Q21 4 19 2 QMOD
Q22 5 25 4a QMOD
Q23 4 21 5 QMOD
Q24 2 23 4a QMOD
*
Q31 6 27 5 QMOD
Q32 7 33 6a QMOD
Q33 6 29 7 QMOD
Q34 5 31 6a QMOD
*
Q41 8 35 7 QMOD
Q42 0 41 8a QMOD
Q43 8 37 0 QMOD
Q44 7 39 8a QMOD
*
.MODEL QMOD NPN (IS=6.83E-14 BF=13 CJE=1pF CJC=607.3PF TF=26.5NS)
R 3 0 1000
.TRAN 0.1US 60MS 0MS 0.1e-4 ;Transient Analysis
.FOUR 50Hz 25 V(3,0)
.OPTIONS ABSTOL=1uA CHGTOL=0.01nC ITL2=100 ITL4=150 RELTOL=0.1
VN1TOL=0.1
.PROBE
.END
```

Tegangan AC yang dibangkitkan oleh inverter 9-level hasil simulasi menggunakan PSPICE ditunjukkan dalam gambar 12. Sedangkan hasil analisis *Fast Fourier Transform* (FFT) [24] ditunjukkan dalam gambar 13.



Gambar 12 Tegangan AC yang dihasilkan inverter 9-level



Gambar 13 Analisis FFT tegangan AC yang dihasilkan inverter 9-level

Nilai tegangan AC fundamental $V_1 = M \times V_{DC} = 3,2 \times 100 = 320$ Volt. Berdasarkan analisis FFT dalam gambar 13, komponen harmonis ke-3 (150Hz), ke-5 (250Hz) dan ke-7 (350Hz) tereliminasi sama sekali. Sedangkan harmonis

tingkat-tinggi yang masih tersisa adalah: $V_9 = 11,7328$ Volt, $V_{11} = 14,2621$ Volt, $V_{13} = 14,2989$ Volt, $V_{15} = 3,0544$ Volt, $V_{17} = 10,6403$ Volt, $V_{19} = 13,0515$ Volt, $V_{21} = 1,0976$ Volt.

Setiap komponen harmonisa tingkat-tinggi semuanya bernilai kurang dari 5%. Parameter untuk menentukan jumlah keseluruhan harmonisa disebut *total harmonic distortion* (THD) [25]. Dari hasil perhitungan dan simulasi, nilai THD dari tegangan AC pada gambar 12 adalah sebesar THD = 9.95%.

IV. KESIMPULAN

Inverter multilevel yang bekerja berdasarkan teknik SHEPWM memerlukan sudut-sudut pensaklaran yang tepat agar harmonisa-harmonisa tertentu yang ditarget tereliminasi sama sekali. Inverter 5-level bertujuan mengeliminasi komponen harmonisa ke-3, inverter 7-level mengeliminasi harmonisa ke-3 dan 5, sedangkan inverter 9-level mengeliminasi harmonisa ke-3, 5 dan, 7. Semakin tinggi level inverter semakin banyak komponen harmonisa yang dapat dieliminasi, namun memerlukan pulsa-pulsa pensaklaran yang semakin banyak pula.

Analisis hasil simulasi inverter multilevel menggunakan perangkat lunak PSPICE mampu menunjukkan keandalan inverter multilevel SHEPWM dalam mengeliminasi harmonisa-harmonisa tegangan AC yang dihasilkan.

REFERENSI

- [1] A. Hiendro, Syaifurrahman, D. Triyanto, and Junaidi, "Solution trajectories for a single-phase programmed PWM inverter," *TELKOMNIKA Indonesian Journal of Electrical Engineering*, vol. 12, no. 7, pp. 5121-5128, 2014.
- [2] B.P. Chandran, A.I. Selvakumar, and F.M.Mathew, "Integrating multilevel converters application on renewable energy sources—A survey," *Journal of Renewable and Sustainable Energy*, vol. 10, no. 6, p. 065502, 2018.
- [3] Y. Lei, C.B. Barth, S. Qin, W.C. Liu, I. Moon, A. Stillwell, D. Chou, T. Foulkes, Z. Ye, Z. Liao, and R.C.N. Pilawa-Podgurski, "A 2-kW single-phase seven-level flying capacitor multilevel inverter with an active energy buffer," *IEEE Transactions on Power Electronics*, vol. 32, no. 11, pp. 8570-8581, 2017.
- [4] R. Castillo, B. Diong, and P. Biggers, "Single-phase hybrid cascaded h-bridge and diode-clamp multilevel inverter with capacitor voltage balancing," *IET Power Electronics*, vol. 11, no.4, pp. 700-707, 2018.
- [5] Y. Suresh, J. Venkataramaiah, A.K. Panda, C. Dhananjayulu, and P. Venugopal, "Investigation on cascade multilevel inverter with symmetric, asymmetric, hybrid and multi-cell configurations," *Ain Shams Engineering Journal*, vol. 8, no. 2, pp. 263-276, 2017.
- [6] H. Akagi, "A review of developments in the family of modular multilevel cascade converters," *IEEJ Transactions on Electrical and Electronic Engineering*, vol. 13, pp. 1222-1235, 2018.
- [7] B.A. Yomahudaya dan T. Sutikno, "Pembangkit sinyal SPWM untuk multilevel inverter satu fasa lima tingkat berbasis mikrokontroler ATmega32," *Jurnal Ilmu Teknik Elektro Komputer dan Informatika (JITEKI)*, vol. 3, no. 2, pp. 73-81, 2017.
- [8] F. Chen and W. Qiao, "A general space vector PWM scheme for multilevel inverters," in *Proc. 2016 IEEE Energy Conversion Congress and Exposition (ECCE)*, Milwaukee, WI, 2016, pp. 1-6, doi: 10.1109/ECCE.2016.7854687.
- [9] S.A. Amamra, K. Meghriche, A. Cherifi, and B. François, "Multilevel inverter topology for renewable energy grid integration," *IEEE Transactions on Industrial Electronics*, vol. 64, no. 11, pp. 8855-8866, 2017.
- [10] M.B. Satti, A. Hasan, and M.I. Ahmad, "A new multilevel inverter topology for grid-connected photovoltaic systems," *International Journal of Photoenergy*, vol. 2018, pp. 1-9, 2018.
- [11] H. Matalata and V. Yusiana, "New topology multilevel inverter type diode clamped five level single phase," *IOP Conference Series: Materials Science and Engineering*, vol. 807, p. 012039, 2020.
- [12] S. Lee and J. Kim, "Optimized modeling and control strategy of the single-phase photovoltaic grid-connected cascaded H-bridge multilevel inverter," *Electronics*, vol. 7, no.9, pp. 1-16, 2018.
- [13] X. Tang, J. Zhang, Z. Liu, and M. Zhang, "A switching frequency optimized space vector pulse width modulation (SVPWM) scheme for cascaded multilevel inverters," *Energies*, vol. 10, no. 5, pp. 1-18, 2017.
- [14] R. Ni, M. Sha, J.X. Guang, Y. Ding, and J. Chen, "Comparison of harmonics between SVPWM and SHEPWM," in *Proc. The 3rd International Conference on Electrical and Information Technologies for Rail Transportation (EITRT)*, 2017, pp. 491-499, DOI: 10.1007/978-981-10-7986-3_51.
- [15] Z. Gong, Q. Cui, X. Zheng, P. Dai, and R. Zhu, "An improved imperialist competitive algorithm to solve the selected harmonic elimination pulse-width modulation in multilevel converters," *Energies*, vol. 11, no. 11, pp. 1-16, 2018.
- [16] K.H. Law, W. P. Q. Ng, and W. K. Wong, "Flyback cascaded multilevel inverter Based SHE-PWM control for STATCOM applications," *International Journal of Power Electronics and Drive System (IJPEDS)*, vol. 8, no. 1, pp. 100-108, 2017.
- [17] P.L. Kamani and M.A. Mulla, "Univariate equation-based SHE-PWM for multilevel inverter," *Sādhanā*, vol. 44, no. 5, pp. 1-5, 2019.
- [18] M. Salman, I. Ul Haq, T. Ahmad, H. Ali, A. Qamar, A. Basit, M. Khan, and J. Iqbal, "Minimization of total harmonic distortions of cascaded H-bridge multilevel inverter by utilizing bio inspired AI algorithm," *EURASIP Journal on Wireless Communications and Networking*, vol. 66, pp. 1-12, 2020.
- [19] M.S.A. Dahidah, G. Konstantinou, and V.G. Agelidis, "A review of multilevel selective harmonic elimination PWM: Formulations, solving algorithms, implementation and applications," *IEEE Transactions on Power Electronics*, vol.30, no. 8, pp.4091-4106, 2015.
- [20] S. Ahmad, M. Meraj, A. Iqbal, and I. Ashraf, "Selective harmonics elimination in multilevel inverter by a derivative-free iterative method under varying voltage condition," *ISA Transaction*, vol. 92, pp. 241-256, 2019.
- [21] H. Behbahanifard, S. Abazari, and A. Sadoughi, "New scheme of SHE-PWM technique for cascade multilevel inverters with regulation of DC voltage sources," *ISA Transaction*, vol. 97, pp. 44-52, 2020.
- [22] A. Hiendro, I. Yusuf, Junaidi, T.P. Wigyarianto, and Y.M. Simanjuntak, "Optimization of SHEPWM cascaded multilevel inverter switching patterns," *International Journal of Power Electronics and Drive System (IJPEDS)*, vol. 11, no. 3, pp. 1570-1578, 2020.
- [23] H. Gunawan, *Analisis Fourier dan Wavelet*, cetakan pertama, Bandung, Indonesia: Fakultas Matematika dan Ilmu Pengetahuan Alam ITB, 2017.
- [24] M. Forman, "A fast fourier transform technique and its application to fourier spectroscopy," *Journal de Physique Colloques*, vol. 28, no. C2, pp. C2-58-C2-61, 1967.
- [25] N. Prabaharan and K. Palanisamy, "A comprehensive review on reduced switch multilevel inverter topologies, modulation techniques and applications," *Renewable and Sustainable Energy Reviews*, vol. 76, pp. 1248-1282, 2017.